code:
;-------------------------------------------------
; 廁所ハプニングイベント
;-------------------------------------------------
@EVENT_TOILET_HAPPENING
#DIM TGT_MEMO

	; TARGETの退避
	TGT_MEMO = TARGET

	; 発生確認＆対象の選別
	CALL EVENT_TOILET_HAPPENING_ENABLE
	TARGET = RESULT

	; ふさわしい対象が居る場合は廁所イベント本体発生
	SIF RESULT
		CALL EVENT_TOILET_HAPPENING_MAIN

	; TARGETの復帰
	TARGET = TGT_MEMO

RETURN 1

;-------------------------------------------------
; 廁所イベント 発生確認
;   返り値が 0 > のとき、イベント発生
;   その場合、返り値が発生対象の番号となる
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_ENABLE
#DIM HEROINE
	; 自宅でのみ発生
	SIF !PLACE("自宅")
		RETURN 0

	; 你尿意確認
	;   ちょっと緩めに判定
	SIF BASE:PLAYER:尿意 < 3000
		RETURN 0

	; 既に発生済みの場合発生しない
	; 発生は１日１回のみ
	SIF CEVENT("廁所ハプニング", PLAYER)
		RETURN 0

	; 対象の選定
	HEROINE = FIND_COND("廁所イベント可能")
	IF HEROINE
		; ヒロインの尿意判定
		SIF BASE:HEROINE:尿意 < 3000
			HEROINE = 0

		; 条件を満たしたうえで随机判定
		SIF RAND:10 > 3
			HEROINE = 0
	ENDIF

	RETURN HEROINE

RETURN 1

;-------------------------------------------------
; 廁所イベント メイン処理
;   TARGET は対象となるヒロインに変更済み
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_MAIN
#DIMS EVENTS
#DIMS MSGS
#DIM SELNUM

	EVENTS =

	; 描写１
	DRAWLINE
	PRINTFORMW %TEXTR("偶然/突然间/＿")%有了尿意的%CALLNAME:PLAYER%面向%TEXTR("洗手间/卫生间")%。
	PRINTL
	PRINTW
	PRINTW 乱七八糟
	PRINTL
	PRINTW
	PRINTFORMW 打開厕所的門、%CALLNAME:PLAYER%看到了%CALLNAME:TARGET%%TEXTR("呆然と/用吃惊的表情/呆気的様子")%看着%CALLNAME:PLAYER%……
	PRINTFORMW %CALLNAME:TARGET%在%TEXTR("厕所/＿")%的马桶上就像是%TEXTR("坐下/放下腰")%%TEXTR("办着事/排泄着")%。
	; 下腹部が見える格好
	IF CHECK_SKIRT("裙") == 0
		IF CHECK_CLO("外套露出")
			PRINTFORM %OUTER(TARGET, "略")%被敞開、
		ELSE
			PRINTFORM 看到的不是裙子、
		ENDIF
		PRINTFORMW 从暴露出的%CALLNAME:TARGET%的%TEXTR("会阴/秘所/陰部")%里、淡黄色的%TEXTR("小水/尿/圣水")%一边描画缓慢的拱形一边落下到马桶发出了水声。
	ELSE
		PRINTFORMW 与一点点的水音一起漂浮着微弱的氨气味。
	ENDIF

	; 口上呼び出し(メイン)
	CALL KOJO_DAILYCOM, "廁所ハプニング"

	; ヒロインから招かれる場合
	IF TALENT:淫乱 + TALENT:好色 && COND("欲求不満") >= 2 && BASE:欲求不満/10 > RAND:10
		SELNUM = 99
	ELSE
		; 対応
		EVENTS = 関上門
		SIF TALENT:淫乱 || MARK:屈服刻印 >= 3 || COND("合意：检查阴部")
			EVENTS += "/就这様继续看"
		CALL PRINT_SELECT, EVENTS
		SELNUM = RESULT
	ENDIF

	; 対応後の処理
	SELECTCASE SELNUM
		CASE 0
		;   ドアを閉める
			PRINTFORMW %CALLNAME:PLAYER%呆呆地望着眼前的情景、慌忙関上了門。
			PRINTFORMW …………
			PRINTFORMW ……
			PRINTFORMW …………隔着淡淡的門传来微弱的水声……
			PRINTFORMW ……
			PRINTFORMW
			PRINTFORM 不久%TEXTR("办完事/从厕所里出来")%的%CALLNAME:TARGET%出现了
			MSGS = 脸上腾起绯红/一直红到耳朵根
			SIF !TALENT:恋慕 && !TALENT:淫乱 && ABL:露出癖 < 3
				MSGS += @"\/看了一眼想要说什么的%CALLNAME:PLAYER%"
			PRINTFORM %TEXTR(MSGS)%
			PRINTFORMW %TEXTR("装做冷静的/若无其事的")%离開了……

		CASE 1
			;  眺める
			PRINTFORM 因为打算%TEXTR("带走/无视/制止")%慌张的%CALLNAME:TARGET%而関上了厕所的門、察觉到%CALLNAME:PLAYER%意図的%CALLNAME:TARGET%
			IF TALENT:不知羞恥
				MSGS = 并不是什么特别害羞的事情/像是要展示一様
				SIF TALENT:強気
					MSGS += "/自己大大地/一边对上視線一边"
				SIF TALENT:開朗
					MSGS += "/很轻松地/很開心地/姿势非常不雅地/一边突出着自己的腰一边"
				SIF TALENT:高慢
					MSGS += "/以好像是要把自信溢出来的態度/以充满優越感的态度"
				PRINTFORM %TEXTR(MSGS)%
			ELSE
				PRINTFORM %TEXTR("害羞/观念转变/似乎放弃了")%地%TEXTR("垂下眼睛/背对着脸")%
			ENDIF
			SIF CHECK_SKIRT("裙")
				PRINTFORM 在把裙子打褶的同時
			;低確率でM字開脚姿勢
			IF (TALENT:淫乱 || ABL:露出癖 >= 5) && RAND:4 == 0
				PRINTFORM 大腿像M字開脚一様展開、
			ELSE
				PRINTFORM 打開着腿
			ENDIF
			PRINTFORMW %TEXTR("黄色的液体/带着微弱氨臭的液体/尿液/聖水")%从尿道给%TEXTR("迸出来/流出来")%的様子……
			PRINTFORMW …………
			PRINTFORMW ……
			PRINTFORMW
			CALL EVENT_TOILET_HAPPENING_EX("見せつけくぱぁ")
			IF RESULT
				PRINTFORM %CALLNAME:PLAYER%的视线
				IF MARK:屈服刻印 >= 3 || ABL:受虐属性 >= 4
					PRINTFORM %TEXTR("腼腆地笑了笑/害羞地接受了")%
				ELSE
					PRINTFORM %TEXTR("脸变成了红色/带着忍受羞耻感的表情接受了考验")%
				ENDIF
				PRINTFORM %CALLNAME:TARGET%小心翼翼地擦干%TEXTR("膣穴/私密处/尿尿的地方")%、收拾完毕、撇下%CALLNAME:PLAYER%%TEXTR("仓促的/害羞的")%离開了厕所……
			ELSE
				PRINTFORMW 不久、结束了%TEXTR("排泄/排尿")%的%CALLNAME:TARGET%以%TEXTR("不満/不知道该说什么")%的脸色很快地结束了善后、%TEXTR("仓促的/逃似的")%离開了%TEXTR("厕所/洗手间")%。
			ENDIF

		CASEELSE
			; 誘われた
			PRINTFORM %CALLNAME:TARGET%%TEXTR("微笑着/笑脸浮现着")%、%TEXTR("突然间/不经意间")%的%TEXTR("小把戏/恶作剧")%、意料之外的事态让%CALLNAME:PLAYER%发呆、
			PRINTFORMW %TEXTR("坐在马桶上/维持着上厕所的姿勢")%就那様邀请%CALLNAME:PLAYER%进厕所……
			PRINTFORM %CALLNAME:PLAYER%一进厕所、%CALLNAME:TARGET%就说
			SIF CHECK_SKIRT("裙")
				PRINTFORM 在把裙子打褶的同時
			PRINTFORMW 打開胯股、用手指展開%TEXTR("阴唇/女陰/女性器")%从尿道那里%TEXTR("小水/尿液/黄金水/聖水")%%TEXTR("迸发/流出/飛散")%的様子被看到了。
			PRINTFORMW 与氨%TEXTS("气")%排出一起发出%TEXTR("响亮/大的")%水声的%TEXTR("小水/尿液/黄金水/聖水")%被便器吸入了……
			PRINTFORMW …………
			PRINTFORMW ……
			PRINTFORMW
			CALL EVENT_TOILET_HAPPENING_EX("Ｈイベント")
			IF RESULT
				; 派生イベントが実行された
				PRINTFORMW %CALLNAME:PLAYER%和%CALLNAME:TARGET%一起整理了厕所里残留的性行为的痕迹、好像什么都没发生过一様离開了厕所……
			ELSE
				PRINTFORMW 不久结束了%TEXTR("排泄/排尿")%的%CALLNAME:TARGET%以%TEXTR("满足/满意/開心/轻松")%脸色结束了善后、与%CALLNAME:PLAYER%一起离開了%TEXTR("厕所/手洗间")%。
			ENDIF
	ENDSELECT

	;ドアを即閉じた以外だと露出経験取得
	IF SELNUM
		; 経験取得
		EXP:露出経験 += 1
		PRINTL
		PRINTFORMW 曝光经验+1

		;ちんこを弄脏的ままにしておくのはあまり良くないので綺麗にする
		SIF PENIS(MASTER) && COND("Ｐ汚れ", MASTER)
			CALL RESET_STAIN_EX, "Ｐ", MASTER
	ENDIF
	DRAWLINE

	; 尿意処理
	CALL SETFLAG_DAILY, "廁所", TARGET
	CALL SETFLAG_DAILY, "廁所", PLAYER

	;廁所ハプニング発生フラグON
	CALL SET_CEVENT, "廁所ハプニング", PLAYER

RETURN 1

;-------------------------------------------------
; 廁所ハプニング 派生イベント
;   一定条件を満たしたときに発生する派生イベント
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_EX(ARGS)
	; 発生確認
	SIF EVENT_TOILET_HAPPENING_EX_ENABLE(ARGS) == 0
		RETURN 0

	; メイン処理
	CALL EVENT_TOILET_HAPPENING_EX_MAIN(ARGS)

	RETURN 1

;-------------------------------------------------
; 廁所ハプニング 派生イベント 続行判断
;   ※拘るならば要調整
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_EX_ENABLE(ARGS)
#FUNCTION
#DIMS TABLE

	VARSET TABLE

	SELECTCASE ARGS
		CASE "見せつけくぱぁ"
			SIF COND("胎内の精液量") < 40000
				RETURNF 0
			RETURNF 1
		CASE "Ｈイベント"
			;どれもチンコが必要
			SIF PENIS(MASTER) == 0
				RETURNF 0

			;イベントが発生するか？
			SIF CANTRAIN("自慰") && ABL:露出癖 >= 3 && ABL:自慰中毒 >= 3
				TABLE += "相互自慰/"
			SIF CANTRAIN("手淫")
				TABLE += "手淫/"
			SIF CANTRAIN("口交")
				TABLE += "口交/"
			SIF CANTRAIN("正常位") || CANTRAIN("後背位")
				TABLE += "Ｖ性交/"
			SIF CANTRAIN("Ａ正常位")
				TABLE += "Ａ性交/"
			SIF TABLE == ""
				RETURNF 0

			; 主动誘っているので高確率
			SIF RAND:10 >= 8
				RETURNF 1
		CASEELSE
			PRINTFORM 错误！%ARGS%未定义
	ENDSELECT

	; 性処理合意がある場合は高確率で続行
	SIF COND("合意：性処理") && RAND:10 >= 3
		RETURNF 1
	; 淫乱判定
	SIF TALENT:淫乱 && RAND:10 >= 5
		RETURNF 1

	; 性処理合意も、淫乱もない場合、恋慕がなければ意向の判定は無効
	SIF !TALENT:恋慕
		RETURNF 0

	; TODO
	;  追加条件

	RETURNF 0

;-------------------------------------------------
; 廁所ハプニング 派生イベント メイン処理
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_EX_MAIN(ARGS)
#DIMS TABLE
#DIMS EVENTS

	; テーブル初期化
	VARSET EVENTS
	VARSET TABLE
	;ここで起きうるイベントのSELECTCOMが必要なら決めてしまう
	SELECTCASE ARGS
		CASE "見せつけくぱぁ"
			EVENTS = 見せつけくぱぁ
			SELECTCOM = GETNUM(TRAINNAME, "扒開阴道")
		CASE "Ｈイベント"
			SIF CANTRAIN("自慰") && ABL:露出癖 >= 3 && ABL:自慰中毒 >= 3
				TABLE += "相互自慰/"
			SIF CANTRAIN("手淫")
				TABLE += "手淫/"
			SIF CANTRAIN("口交")
				TABLE += "口交/"
			SIF CANTRAIN("正常位") || CANTRAIN("後背位")
				TABLE += "Ｖ性交/"
			SIF CANTRAIN("Ａ正常位")
				TABLE += "Ａ性交/"

			EVENTS = %TEXTR(TABLE)%

			SELECTCASE EVENTS
				CASE "相互自慰"
					SELECTCOM = GETNUM(TRAINNAME, "自慰")
				CASE "手淫"
					SELECTCOM = GETNUM(TRAINNAME, "手淫")
				CASE "口交"
					SELECTCOM = GETNUM(TRAINNAME, "口交")
				CASE "Ｖ性交"
					SIF CANTRAIN("正常位")
						SELECTCOM = GETNUM(TRAINNAME, "正常位")
					SIF CANTRAIN("後背位") && (CANTRAIN("正常位") == 0 || RAND:2)
						SELECTCOM = GETNUM(TRAINNAME, "後背位")
				CASE "Ａ性交"
					SELECTCOM = GETNUM(TRAINNAME, "肛門後背位")
			ENDSELECT
	ENDSELECT

	STR:日常イベント名 = 廁所イベント：%EVENTS%

	SELECTCASE EVENTS
		CASE "見せつけくぱぁ"
			PRINTFORMW 不久、水音变弱而中断了、%CALLNAME:TARGET%却以别扭的姿态扭捏着什么……
			PRINTFORM 等了一会儿、脸红了的%CALLNAME:TARGET%
			PRINTFORM %TEXTR("双腿像展示一样以更大幅度打開/身體後仰將腰肢挺起")%
			PRINTFORMW 用双手的手指展開%TEXTR("自己的秘所/阴唇")%……
			PRINTFORM 水滴残留的女阴小穴蠢蠢欲动、从孔穴深处传来
			IF CFLAG:Ｖ已中出
				PRINTFORM %TEXTR("刚才/刚刚")%
			ELSE
				PRINTFORM %TEXTR("上次/前回")%
			ENDIF
			PRINTFORM %TEXTR("充分的/大量得")%%TEXTR("注込/中出")%
			IF CFLAG:Ｖ已中出
				PRINTFORM %TEXTR("新鮮的/鲜活的")%
			ELSE
				PRINTFORM %TEXTR("留在了子宫里/白濁液")%
			ENDIF
			PRINTFORMW %TEXTR("精液/白濁液")%发出黏腻的声音溢了出来……

			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

		CASE "相互自慰"
			PRINTFORM 看着%CALLNAME:TARGET%的%TEXTR("排尿姿势/羞耻姿势/春姿")%的%CALLNAME:PLAYER%、取出%TEXTR("硬邦邦/臨戦態勢/勃起/元気满满")%的%TEXTR("小兄弟/丁丁/自己的分身")%、
			PRINTFORM 将那个
			IF TALENT:淫乱
				PRINTFORM 凝视%TEXTR("充满欲望的双眼/期待的眼神/热情满满的瞳孔")%
			ELSE
				PRINTFORM 偷看%TEXTR("一晃一晃/背对着脸")%
			ENDIF
			PRINTFORMW 在%CALLNAME:TARGET%面前%TEXTR("安慰自己/自慰")%。
			PRINTFORM %CALLNAME:TARGET%就坐在马桶上
			IF TALENT:淫乱
				PRINTFORM %TEXTR("吮吸着自己的食指/脸颊写满了欲望/瞳孔潤湿了")%
			ELSE
				PRINTFORM %TEXTR("难为情的/羞耻的")%
			ENDIF
			PRINTFORMW %CALLNAME:PLAYER%进行着%TEXTR("自慰/一人游戏")%、%TEXTR("観察/看見/视线離不開")%%CALLNAME:TARGET%用自己的手指剥開%TEXTR("阴唇/女性器/女陰")%伸进去进行着%TEXTR("自慰/一人游戏")%……

			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

			PRINTFORMW 吐出甜蜜的呼吸、%CALLNAME:TARGET%的身体微微颤抖、
			PRINTFORM %CALLNAME:PLAYER%看着只是%TEXTR("自慰/用自己的手指/一人游戏就")%
			IF NOWEX:Ｃ絶頂 + NOWEX:Ｖ絶頂 + NOWEX:Ａ絶頂 + NOWEX:Ｂ絶頂 + NOWEX:尿道絶頂
				PRINTFORMW 绝顶了的%CALLNAME:TARGET%的脸射出了精液。
			ELSE
				PRINTFORMW 就露出了開心的表情的%CALLNAME:TARGET%的脸射出了精液。
			ENDIF
			PRINTFORMW 面对着%CALLNAME:PLAYER%的%TEXTR("精液/白浊")%在脸上%TEXTR("大量射出/滴滴荟萃/狠狠地颜射")%、%CALLNAME:TARGET%一脸茫然……
		CASE "手淫"
			PRINTFORM 突然停止了水声、%TEXTR("结束了排尿/结束排泄的/因为排尿的快感而阵阵抖起身体的")%%CALLNAME:TARGET%
			IF TALENT:害羞 || ABL:露出癖 < 5
				PRINTFORM 将%TEXTR("满是羞耻的/像是害羞地烧起来了的")%脸%TEXTR("背过去/转过去/低下来")%忍耐着内心羞耻到不行的痛苦…
			ELSE
				PRINTFORM 没有任何想要隐藏起自己%TEXTR("似乎是因为被看到而兴奋了起来而/因为满足了下流的欲望而")%露出的下流高潮的表情的样子。
			ENDIF
			PRINTFORMW 看到了那样的%CALLNAME:TARGET%而%TEXTR("膨胀起来的/勃起的/变硬的")%%CALLNAME:MASTER%的%TEXTR("阴茎/欲望/男性器/肉棒")%出现在了%CALLNAME:TARGET%的面前。
			PRINTFORM %TEXTR("用满是羞耻的脸颊/用露出了焦躁的表情/用湿润了的双眼/慢慢地吐出湿润的吐息")%
			IF TALENT:淫乱 && RAND:3
				PRINTFORM 绽放出淫乱的%TEXTR("笑容/笑颜/神色")%的%CALLNAME:TARGET%一句话也没有说
			ELSEIF (TALENT:小悪魔 || ABL:施虐属性 >= 3) && RAND:3
				PRINTFORM %TEXTR("露出了轻薄的笑容/带着优越感放松了表情/嘲笑着那样不堪的丑态")%、像是准备惩罚一样靠近了%CALLNAME:MASTER%的%CALLNAME:TARGET%
			ELSEIF TALENT:害羞 && RAND:3
				PRINTFORM 露出了似乎是在说着「%TEXTR("这样下去该怎么办好/是自己的错吗/好像很辛苦的样子……但是")%」的表情的%CALLNAME:TARGET%
			ELSE
				PRINTFORM %TEXTR("咕噜地咽下了一口唾沫的/苦笑着的/微微低下頭的")%%CALLNAME:TARGET%
			ENDIF
			IF (TALENT:恋慕 && ABL:技巧 >= 5) || TALENT:淫乱
				PRINTFORMW 熟练地只用嘴唇将%CALLNAME:MASTER%的%TEXTR("凶猛阴茎/欲望/燥熱/怒張/肉棒/阴茎")%从内裤中解放出来、毫不犹豫地%TEXTR("帮助处理/用手指触摸着開始温柔的奉仕/用手指抚摸着湿润的前端")%……
			ELSEIF ABL:技巧 >= 3
				PRINTFORMW %TEXTR("顺从地/单手抓着/什么都没有说地")%拉下%CALLNAME:MASTER%的内裤、将自己的手指用唾液润滑之后開始了%TEXTR("撸动/奉仕")%……
			ELSE
				PRINTFORMW 将手放在%CALLNAME:MASTER%的内裤上、%TEXTR("艰难地/笨手笨脚地/犹豫不决地")%%TEXTR("開始了手淫/開始用手奉仕/握住了那根冒出来的东西")%……
			ENDIF
			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

		CASE "口交"
			PRINTFORM %CALLNAME:PLAYER%的视线
			IF TALENT:恋慕 || TALENT:淫乱
				PRINTFORM 一边沐浴在
			ELSE
				PRINTFORM 一边忍耐
			ENDIF
			PRINTFORMW 看着%TEXTR("不断地撒着尿/持续排泄/聖水排出")%的%CALLNAME:TARGET%的%TEXTR("私密/羞耻")%姿势、涌出了与尿意不同的排泄欲……
			PRINTFORM 取出%TEXTR("勃起/硬邦邦/臨戦態勢")%的%TEXTR("丁丁/男性器/小兄弟")%后、%CALLNAME:TARGET%说
			IF TALENT:恋慕 || TALENT:淫乱
				PRINTFORM 似乎在焦急地等待
			ELSE
				PRINTFORM 犹豫不决
			ENDIF
			PRINTFORMW 把%CALLNAME:PLAYER%的那个%TEXTR("含在嘴里/開始口交/吞到最里面")%……

			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

			; さらに派生(ToDo)

			PRINTFORM 就算从%TEXTR("秘所/女陰/尿道口")%中溢出的%TEXTR("小水/尿液/聖水")%停了下来、%CALLNAME:TARGET%也不会停止%TEXTR("口交/舔舐")%阴茎的、
			PRINTFORMW 在%CALLNAME:TARGET%的%TEXTR("口中/舌上/喉咙深处")%射出大量的精液、一边咳嗽一边将嘴里积存的%CALLNAME:PLAYER%的精液一滴不剩地吞下。

		CASE "Ｖ性交"
			; 避孕套の有無
			TEQUIP:PLAYER:避孕套 = 1
			SIF COND("女性主導生性交")
				TEQUIP:PLAYER:避孕套 = 0

			IF ABL:受虐属性 >= 3 || ABL:露出癖 >= 5
				PRINTFORM 为了展现自己的排泄姿态而张開双脚
			ELSE
				PRINTFORM 一边忍受着害羞
			ENDIF
			PRINTFORMW 持续排泄的%CALLNAME:TARGET%的姿态让情欲沸腾……
			PRINTFORM 暂且把%TEXTR("勃起/硬邦邦/準備万端")%的%TEXTR("阴茎/男性器")%交给%TEXTR("上完厕所/排泄終止/尿液途中流尽/聖水排完")%的%CALLNAME:TARGET%
			PRINTFORM 在%CALLNAME:TARGET%眼前、%CALLNAME:TARGET%是
			IF TEQUIP:PLAYER:避孕套
				SELECTCASE RAND:2
					CASE 0
						PRINTFORM %SKIRT(TARGET, "略")%的口袋
					CASE 1
						PRINTFORM 在备置的小盒子里
				ENDSELECT
				PRINTFORM 从中取出避孕套
				SIF ABL:奉仕精神 > 3
					PRINTFORM 用嘴巴
				PRINTFORM 快速安装、
			ENDIF
			; 体位が後背位
			IF SELECTCOM == GETNUM(TRAINNAME, "後背位")
				PRINTFORM 屁股朝这边坐上马桶
			ELSE
				PRINTFORM 像抱着自己的脚一様
			ENDIF
			PRINTFORM 采取姿势
			IF MARK:屈服刻印 > 1
				PRINTFORM 充满期待的眼睛
			ELSE
				PRINTFORM 害羞地红着脸
			ENDIF
			PRINTFORMW %CALLNAME:PLAYER%的%TEXTR("阴茎/肉棒")%将%TEXTR("私密深处/魅惑的柔肉")%%TEXTR("完全吞没/深深插入")%……

			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

			; さらに派生(ToDo)

		CASE "Ａ性交"
			PRINTFORM %TEXTR("排泄終止/上完厕所/聖水排出完毕")%的%CALLNAME:TARGET%
			IF MARK:屈服刻印 >= 3 || ABL:受虐属性 >= 4
				PRINTFORM 带着满怀期待的眼神
			ELSE
				PRINTFORM 脸红到耳朵
			ENDIF
			PRINTFORMW 手扶着马桶、像是在强调%TEXTR("菊花/肛門")%一様伸出屁股。
			PRINTFORMW 响应着%TEXTR("欲望/希望/诱惑/期待")%、将阴茎塞进%TEXTR("突然缩了一下/紧紧绷住")%的%CALLNAME:TARGET%后面的洞穴、%CALLNAME:TARGET%发出像被拔掉了什么的长长的叹息、一口气吞噬了%CALLNAME:PLAYER%的%TEXTR("陰茎/分身/肉棒")%……

			; 仮想指令実行処理
			CALL EVENT_TOILET_HAPPENING_EX_TRAIN_SIM, EVENTS

		CASEELSE
			PRINTFORMW 错误！%EVENTS%未定义的内容。
			RETURN 0
	ENDSELECT

;-------------------------------------------------
; 廁所ハプニング 派生イベント 指令仮想実行処理
;-------------------------------------------------
@EVENT_TOILET_HAPPENING_EX_TRAIN_SIM(ARGS)
#DIM LCOUNT
#DIM LINE_MEMO

	;絶頂度合いを重置
	VARSET EX
	VARSET NOWEX
	VARSET EX:PLAYER:0
	VARSET NOWEX:PLAYER:0
	VARSET LOCAL

	;前準備開始
	CALL SIMULATE_TRAIN_FIRST

	SELECTCASE ARGS
		CASE "相互自慰"
			; 相互 なのでヒロイン側の絶頂処理を先に済ませる
			FOR LCOUNT, 0, MAX(MIN(ABL:欲望, 10)/2, MIN(ABL:自慰中毒, 5), 1) + RAND:3
				CALL SIMULATE_TRAIN_COM, SELECTCOM, 1

				;絶頂済みならここでおしまい
				SIF NOWEX:Ｃ絶頂 + NOWEX:Ｖ絶頂 + NOWEX:Ａ絶頂 + NOWEX:Ｂ絶頂 + NOWEX:尿道絶頂
					BREAK
			NEXT
			;PRINTFORMW {EXP_GAIN("Ｖ経験")}

			SELECTCOM = GETNUM(TRAINNAME, "自慰")
		CASE "手淫"

		CASE "口交"

		CASE "Ｖ性交"

		CASE "Ａ性交"

		CASE "見せつけくぱぁ"
			; 絶頂とは関係ない
			LOCAL = 1
		;CASE "口交後飲尿"
		;	SELECTCOM = GETNUM(TRAINNAME, "扒開阴道")
		;	; 絶頂とは関係ない
		;	LOCAL = 1
	ENDSELECT

	; COM格子
	IF SELECTCOM < 0
		PRINTFORMW 错误！对应%ARGS%的COM不存在({SELECTCOM})
		RETURN 0
	ENDIF

	SIF !LOCAL
		LOCAL = 999

	; 実行ループ
	FOR LCOUNT, 0, LOCAL
		CALL SIMULATE_TRAIN_COM, SELECTCOM, 1

		;射精済みならここでおしまい
		SIF NOWEX:PLAYER:射精
			BREAK
	NEXT

	;経験や珠などの獲得量計算
	CALL SIMULATE_TRAIN_CALC

	;結果発表
	CALL SIMULATE_TRAIN_SHOW

	; 口上対応
	LINE_MEMO = LINECOUNT
	CALL KOJO_DAILYCOM, "廁所Ｈ"
	SIF NOWEX:PLAYER:射精 && LINECOUNT == LINE_MEMO
		CALL SAMEN_SHOOT_MESSAGE

	CALL SETFLAG, "調教外Ｈ終了", MASTER
	CALL SETFLAG, "調教外Ｈ終了", TARGET

;-------------------------------------------------
; 廁所ハプニング イベントフラグ阴蒂ア
;   日期が変わるとき(EVENTTURNEND)に呼ばれる
;-------------------------------------------------
@EVENTTURNEND_TOILET_HAPPENING
CALL DEL_CEVENT, "廁所ハプニング", PLAYER


code:
;-------------------------------------------------
;夜に起こるイベント
;-------------------------------------------------
@EVENT_YOBAI, ARGS
#DIM LCOUNT
;行為の回数
#DIM NUM_PLAY

;CFLAG:一緒に寝た の初期化
CVARSET CFLAG, GETNUM(CFLAG, "一緒に寝た")
;夜袭フラグ
CVARSET CSTR, GETNUM(CSTR, "夜袭")

IF ARGS == ""
	SIF COND("夜袭可能") == 0 || COND("手一杯", MASTER)
		RETURN 0

	SELECTCASE COND("欲求不満")
	CASE 0
		RETURN 0
	CASE 1
		SIF TALENT:淫乱 == 0 && TALENT:恋慕 == 0
			RETURN 0
	ENDSELECT
ENDIF

;夜イベントの決定。内容は CSTR:夜袭 に入っている
CALL PLAY_AT_EVENING
SIF ARGS != ""
	CSTR:夜袭 = %ARGS%
SIF CSTR:夜袭 == ""
	RETURN 0

;夜袭フラグON
SETBIT FLAG:状況判定, 43
TFLAG:颜绘表示中 = 0

;夜袭の衣装に
CALL SET_COORDINATE, "夜袭衣装"

;基本回数は精力依存
NUM_PLAY = LIMIT(BASE:精力/20 + LIMIT(RAND:(BASE:精力/10 + 1), 1, BASE:精力/15), 1, 4)

DRAWLINE
SIF COND("同衾イベント")
	PRINTFORML ＜%CALLNAME:TARGET%的房間＞

SELECTCASE CSTR:夜袭
CASE "夜袭騎乗位"
	CVARSET TEQUIP, GETNUM(TEQUIP, "推倒"), 1

	NUM_PLAY = NUM_PLAY + MIN(3, (ABL:Ｖ感覚 + ABL:精液中毒)/3)

	;PRINTL ＜夜袭＞
	;口上の前後が凝っている角色用
	CALL KOJO_EVENT, "夜袭騎乗位口上操作"
	IF TFLAG:地の文スキップ
		CALL KOJO_EVENT, CSTR:夜袭
	ELSE
		IF COND("同衾イベント")
			PRINTFORMW %CALLNAME:MASTER%刚要躺在床上的时候、%CALLNAME:TARGET%抱了过来
		ELSE
			PRINTFORMW 在%CALLNAME:MASTER%就要上床的时候、突然%CALLNAME:TARGET%进入了房间
		ENDIF

		TFLAG:颜绘表示中 = 0
		CALL EMOTION_PLUS, "躁动难忍", TARGET

		;[高慢][傲慢][坏心眼][強気]持ちなら分岐
		IF TALENT:高慢 || TALENT:傲慢 || TALENT:強気 || TALENT:坏心眼
			PRINTFORMW 「……」
			PRINTFORMW 没有说话、而是用热得湿润的眼睛凝视着%CALLNAME:MASTER%
		;[害羞][清楚]持ちの場合分岐
		ELSEIF TALENT:害羞 || TALENT:清楚
			PRINTFORMW 用充满忧郁和欲望的眼神、視線偷偷上飄窺視著%CALLNAME:MASTER%的情況
			SELECTCASE NO:TARGET
			CASEELSE
				PRINTFORMW 「……」
			ENDSELECT
		;[容易迷恋][献身][小悪魔]持ちなら分岐
		ELSEIF TALENT:容易迷恋 || TALENT:献身 || TALENT:小悪魔
			PRINTFORMW 「……」
			PRINTFORMW 没有说话、而是用热得湿润的眼睛微笑地看着%CALLNAME:MASTER%
		ENDIF
		PRINTFORMW 就这様彼女将%CALLNAME:MASTER%推倒、用蜜壷吞下阴茎、用腰肢激烈地擊打著開始了

		CALL KOJO_EVENT, CSTR:夜袭

		;[高慢][傲慢][坏心眼][強気]持ちなら分岐
		IF TALENT:高慢 || TALENT:傲慢 || TALENT:強気 || TALENT:坏心眼
			PRINTFORMW %CALLNAME:TARGET%的嘴里夹杂着嬌声和悪语的同时、不停淫靡摆动着腰肢来抚慰子宮的干渇…
		;[害羞][清楚]持ちの場合分岐
		ELSEIF TALENT:害羞 || TALENT:清楚
			PRINTFORML %CALLNAME:TARGET%不停地小声编织着歓喜的悲鳴和謝罪话语的同时、
			PRINTFORMW 無心地摆动着腰肢来抚慰子宮的干渇…
		;[容易迷恋][献身][小悪魔]持ちなら分岐
		ELSEIF TALENT:容易迷恋 || TALENT:献身 || TALENT:小悪魔
			PRINTFORMW %CALLNAME:TARGET%微笑着、并妖艳地摇动着腰肢抚慰着子宮的干渇…
		;[開朗][好色][接受快感]持ちの場合分岐
		ELSEIF TALENT:開朗 || TALENT:好色 || TALENT:接受快感
			PRINTFORMW %CALLNAME:TARGET%微笑着、并妖艳地摇动着腰肢抚慰着子宮的干渇…
		ELSE
			PRINTFORMW 就像飢饿的野獣貪婪地吞噬着生肉一般、抚慰着子宮的干渇…
		ENDIF
	ENDIF
	CALL SIMULATE_TRAIN, 65, NUM_PLAY

	CALL CALC, "欲求不満解消", TARGET, MIN(NUM_PLAY*25, 75)

	PRINTFORM 不知道多少次絶頂以後、
	IF TALENT:相思相愛
		PRINTFORMW 相思的两人紧紧抱在一起、落入了深深的睡眠…
		;一緒に眠ったフラグ。朝の相手として優先。
		CFLAG:MASTER:一緒に寝た = TARGET
		CFLAG:一緒に寝た = MASTER
	ELSE
		PRINTFORM 实现了思念的%CALLNAME:TARGET%将
		;相手がいっぱいいるときは紳士協定
		IF RAND:CHARANUM >= 5
			PRINTFORMW 身体清理干净、依依不舍地回到了自己的房间…
		ELSE
			PRINTFORMW %CALLNAME:MASTER%抱住、就这様落入了深深的睡眠…
			CFLAG:MASTER:一緒に寝た = TARGET
			CFLAG:一緒に寝た = MASTER
		ENDIF
	ENDIF
	CVARSET TEQUIP, GETNUM(TEQUIP, "推倒")

CASE "夜袭Ａ騎乗位"
	CVARSET TEQUIP, GETNUM(TEQUIP, "推倒"), 1

	NUM_PLAY = NUM_PLAY + MIN(3, (ABL:Ａ感覚 + ABL:精液中毒)/3)

	;PRINTL ＜夜袭＞
	;口上の前後が凝っている角色用
	CALL KOJO_EVENT, "夜袭Ａ騎乗位口上操作"
	IF TFLAG:地の文スキップ
		CALL KOJO_EVENT, CSTR:夜袭
	ELSE
		IF COND("同衾イベント")
			PRINTFORMW %CALLNAME:MASTER%刚要躺在床上的时候、%CALLNAME:TARGET%抱了过来
		ELSE
			PRINTFORMW 在%CALLNAME:MASTER%就要上床的时候、突然%CALLNAME:TARGET%进入了房间
		ENDIF

		CALL EMOTION_PLUS, "躁动难忍", TARGET

		;[高慢][傲慢][坏心眼][強気]持ちなら分岐
		IF TALENT:高慢 || TALENT:傲慢 || TALENT:強気 || TALENT:坏心眼
			PRINTFORMW 「……」
			PRINTFORMW 没有说话、而是用热得湿润的眼睛凝视着%CALLNAME:MASTER%
		;[害羞][清楚]持ちの場合分岐
		ELSEIF TALENT:害羞 || TALENT:清楚
			PRINTFORMW 用充满忧郁和欲望的眼神、視線偷偷上飄窺視著%CALLNAME:MASTER%的情況
			SELECTCASE NO:TARGET
			CASEELSE
				PRINTFORMW 「……」
			ENDSELECT
		;[容易迷恋][献身][小悪魔]持ちなら分岐
		ELSEIF TALENT:容易迷恋 || TALENT:献身 || TALENT:小悪魔
			PRINTFORMW 「……」
			PRINTFORMW 没有说话、而是用热得湿润的眼睛微笑地看着%CALLNAME:MASTER%
		ENDIF
		PRINTFORMW 就这様\@ TALENT:男性 ? %CALLNAME:TARGET% # 彼女 \@将%CALLNAME:MASTER%推倒、不浄的穴吞下阴茎、用腰肢激烈地擊打著開始了

		CALL KOJO_EVENT, CSTR:夜袭

		;[高慢][傲慢][坏心眼][強気]持ちなら分岐
		IF TALENT:高慢 || TALENT:傲慢 || TALENT:強気 || TALENT:坏心眼
			PRINTFORMW %CALLNAME:TARGET%的嘴里夹杂着嬌声和悪语的同时、不停淫靡摆动着腰肢贪享着肛門的肉欲…
		;[害羞][清楚]持ちの場合分岐
		ELSEIF TALENT:害羞 || TALENT:清楚
			PRINTFORML %CALLNAME:TARGET%不停地小声编织着歓喜的悲鳴和謝罪话语的同时、
			PRINTFORMW 無心地摆动着腰肢来满足肛門贪婪的肉欲…
		;[容易迷恋][献身][小悪魔]持ちなら分岐
		ELSEIF TALENT:容易迷恋 || TALENT:献身 || TALENT:小悪魔
			PRINTFORMW %CALLNAME:TARGET%微笑着、并妖艳地摇动着腰肢满足着肛門贪婪的肉欲…
		;[開朗][好色][接受快感]持ちの場合分岐
		ELSEIF TALENT:開朗 || TALENT:好色 || TALENT:接受快感
			PRINTFORMW %CALLNAME:TARGET%眯起眼微笑着、并不停用腰肢擊打著来满足肛門贪婪的肉欲…
		ELSE
			PRINTFORMW 被完全開発的肛門像吸入一样怒張地吞噬着、尽情贪享着肉欲…
		ENDIF
	ENDIF
	CALL SIMULATE_TRAIN, 75, NUM_PLAY

	CALL CALC, "欲求不満解消", TARGET, MIN(NUM_PLAY*25, 75)

	PRINTFORM 不知道多少次絶頂以後、
	IF TALENT:相思相愛
		PRINTFORMW 相思的两人紧紧抱在一起、落入了深深的睡眠…
		;一緒に眠ったフラグ。朝の相手として優先。
		CFLAG:MASTER:一緒に寝た = TARGET
		CFLAG:一緒に寝た = MASTER
	ELSE
		PRINTFORM 实现了思念的%CALLNAME:TARGET%将
		;相手がいっぱいいるときは紳士協定
		IF RAND:CHARANUM >= 5
			PRINTFORMW 身体清理干净、依依不舍地回到了自己的房间…
		ELSE
			PRINTFORMW %CALLNAME:MASTER%抱住、就这様落入了深深的睡眠…
			CFLAG:MASTER:一緒に寝た = TARGET
			CFLAG:一緒に寝た = MASTER
		ENDIF
	ENDIF
	CVARSET TEQUIP, GETNUM(TEQUIP, "推倒")

CASE "添い寝"
	NUM_PLAY = 1

	;口上の前後が凝っている角色用
	CALL KOJO_EVENT, "添い寝口上操作"
	IF TFLAG:地の文スキップ
		CALL KOJO_EVENT, CSTR:夜袭
	ELSE
		CALL EMOTION_PLUS, "害羞", TARGET

		;PRINTL ＜添い寝？＞
		IF COND("同衾イベント")
			PRINTFORMW %CALLNAME:MASTER%躺在床上的时候、穿着睡衣的%CALLNAME:TARGET%悄悄地～爬到了床上
			CALL KOJO_EVENT, CSTR:夜袭
			PRINTFORMW 然后、抱住%CALLNAME:MASTER%露出了満足的表情、就这様睡着了……
		ELSE
			PRINTFORM %CALLNAME:MASTER%躺在床上时、
			IF RAND:3 == 0
				PRINT 抱着枕頭的
			ELSEIF RAND:2
				PRINT 睡衣姿的
			ELSE
				PRINT 穿着怪獣玩偶睡衣的
			ENDIF
			PRINTFORMW %CALLNAME:TARGET%偷偷溜进了房间
			PRINTFORMW 显然的、是想和%CALLNAME:MASTER%一起睡觉…
			PRINTFORMW 小声地叫了下正在装睡的%CALLNAME:MASTER%之后、%CALLNAME:TARGET%悄悄地～爬到了床上
			CALL KOJO_EVENT, CSTR:夜袭
			PRINTFORMW 然后、抱住%CALLNAME:MASTER%露出了満足的表情、就这様睡着了……
		ENDIF
	ENDIF
	CFLAG:MASTER:一緒に寝た = TARGET
	CFLAG:一緒に寝た = MASTER

CASE "夜袭接吻"
	IF TALENT:恋慕 && CFLAG:好感度 >= 3000 && TALENT:蕩唇
		NUM_PLAY = RAND:3 + 1
	ELSEIF TALENT:恋慕 && CFLAG:好感度 >= 3000
		NUM_PLAY = RAND:2 + 1
	ELSE
		NUM_PLAY = 1
	ENDIF

	;口上の前後が凝っている角色用
	CALL KOJO_EVENT, "夜袭接吻口上操作"
	IF TFLAG:地の文スキップ
		CALL KOJO_EVENT, CSTR:夜袭
	ELSE
		CALL EMOTION_PLUS, "害羞", TARGET

		;PRINTL ＜夜袭接吻＞
		IF COND("同衾イベント")
			PRINTFORMW %CALLNAME:MASTER%在床上刚刚睡着、%CALLNAME:TARGET%悄悄地～抱了过来
		ELSE
			PRINTFORMW %CALLNAME:MASTER%躺在床上时、%CALLNAME:TARGET%偷偷溜进了房间
		ENDIF
		PRINTFORMW %CALLNAME:TARGET%窥视着%CALLNAME:MASTER%的寝顔、抑制不住思念奪走了嘴唇
		CALL KOJO_EVENT, CSTR:夜袭
		PRINTFORMW 把%CALLNAME:MASTER%没有醒来当作是好事、一再貪婪地反复亲吻……
	ENDIF
	CALL SIMULATE_TRAIN, 6, NUM_PLAY

	IF COND("同衾イベント") == 0
		PRINTFORML 片刻后结束了着隐秘的行為、回想着嘴唇上残留的感触的
		PRINTFORMW %CALLNAME:TARGET%依依不舍地离開了房间
	ENDIF

	CALL CALC, "欲求不満解消", TARGET, NUM_PLAY*5
ENDSELECT
PRINTL

;夜袭フラグOFF
CLEARBIT FLAG:状況判定, 43

DRAWLINE

;-------------------------------------------------
;夜袭イベントの決定。何をやるかは CSTR:夜袭 で記録する
;-------------------------------------------------
@PLAY_AT_EVENING
#DIM LCOUNT
;起きうるイベントの判定に使う
VARSET STR:選択肢

;実行可能指令の確認
CALL CHECK_COM_ABLE, TARGET

FOR LCOUNT, 0, 10
	SELECTCASE LCOUNT
	CASE 0
		SIF PENIS(PLAYER) == 0 || TALENT:処女 || TALENT:PLAYER:童貞 || CANTRAIN("推倒騎乗位") == 0
			CONTINUE
		SIF COND("女性主導生性交") == 0
			CONTINUE
		STR:選択肢 += "夜袭騎乗位/"
	CASE 1
		SIF PENIS(PLAYER) == 0 || EXP:PLAYER:Ａ挿入経験 == 0 || CANTRAIN("推倒肛門騎乗位") == 0
			CONTINUE
		STR:選択肢 += "夜袭Ａ騎乗位/"

	CASE 2
		;欲求不満LV2or同衾イベント発生時には、ここより上の過激なPLAYを優先する
		SIF COND("欲求不満") >= 2 || COND("同衾イベント")
			BREAK

		SIF (COND("欲求不満") || AGE(TARGET) >= 2) && COND("同衾イベント") == 0
			CONTINUE
		STR:選択肢 += "添い寝/"
	CASE 3
		SIF TALENT:恋慕 == 0 || CANTRAIN("接吻") == 0
			CONTINUE
		STR:選択肢 += "夜袭接吻/"
	CASEELSE
		BREAK
	ENDSELECT
NEXT
CSTR:夜袭 = %TEXTR(STR:選択肢)%

RETURN 0

license: unknown


code:
;-------------------------------------------------------------------------------------------------
;助手に性欲処理を頼む
;台詞については TRYCALLFORM KOJO_GUEST_K{NO:ASSI}, LOCALS, ASSI で呼び、無い場合には地の文で補填
;おおまかなフラグについてはCEVENTの"助手で性処理："グループを使用しても良い。これは毎ターン初期化される
;得た経験などについてはEXP_GAINで参照できる
;ただし、主人の経験を参照する場合には他の助手相手のPLAYで経験が増えていることがあるため注意が必要
;一回に付き助手経験が+1されるため、EXP_GAIN("助手経験", ASSI)でこのターンに何回相手したかが分かる（現在は１回で終了）
;同様にEXP_GAINで増えた経験を参照可能
;前回やったことについてはPREVCOMで参照可能。
;阴茎が汚れてるかどうかは CHECK_STAIN(STAIN:PLAYER:Ｐ, "Ｐ恥垢")
;-------------------------------------------------------------------------------------------------
@PLAY_WITH_ASSI
#DIM LCOUNT
#DIM MEMO, 10
#DIM MEMO_LINECOUNT
#DIM MEMO_TARGET
#DIM MEMO_ASSI
#DIMS EVENTNAME
#DIMS MEMO_CLO

;おあずけフラグがたっている？
IF CEVENT("助手で性処理：おあずけ", ASSI)
	;口上があるかどうかを行数で格子
	MEMO_LINECOUNT = LINECOUNT

	CALL KOJO_GUEST, "助手で性処理：おあずけ", ASSI

	;台詞が無いなら汎用イベント（台詞は極力没有で地の文のみとする）
	SIF LINECOUNT == MEMO_LINECOUNT
		PRINTFORMW 続きは%CALLNAME:ASSI%的房間でしましょう
	RETURN 0
ENDIF

;SIMULATE_TRAINを使うためにASSIをTARGETにする
;-------------------------------------------------------------------------------------------------
MEMO_TARGET = TARGET
MEMO_ASSI = ASSI
TARGET = ASSI

CALL SIMULATE_TRAIN_FIRST

TARGET = MEMO_TARGET
ASSI = MEMO_ASSI
;-------------------------------------------------------------------------------------------------

;選択回数によるローテーションにはRANDF(10, EXP_GAIN("助手経験", ASSI) )を用いる：保留
;前回と違う人or時間帯ごとの初回なら各種フラグ初期化
IF ASSI != COND("前Ｔ対象") || EXP_GAIN("助手経験", ASSI) == 0
	PREVCOM = -1
	VARSET TEQUIP:ASSI:0, 0, 0, 100
	VARSET TEQUIP:MASTER:0, 0, 0, 100
	CALL DEL_CEVENT_GROUP, "助手で性処理", ASSI
ENDIF

;着ている服を記憶
MEMO_CLO =
FOR LCOUNT, 40, 50
	MEMO_CLO = %MEMO_CLO%{EQUIP:ASSI:LCOUNT},%CSTR:ASSI:LCOUNT%,
NEXT
;確認用
;PRINTFORMW %MEMO_CLO%

;１ターン限定のフラグを消去
CALL DEL_CEVENT_GROUP, "助手で性処理", ASSI
SELECTCOM:5 = -1

;指定が無いなら、やることは口交or舐陰
IF PENIS(MASTER) && RANDIF(1, TALENT:MASTER:男性 == 0)
	SELECTCOM = 91
ELSE
	SELECTCOM = 98
ENDIF

PRINTL

;口上があるかどうかを行数で格子
MEMO_LINECOUNT = LINECOUNT

;内部で好きにイベントとSELECTCOMを決定
CALL KOJO_GUEST, "助手で性処理", ASSI

;台詞が無いなら汎用イベント（台詞は極力没有で地の文のみとする）
IF LINECOUNT == MEMO_LINECOUNT
	SELECTCASE SELECTCOM
	;口交
	CASE 91
		;六九式
		IF ABL:ASSI:施虐属性 && ABL:MASTER:受虐属性 && CHECK_CLO("胖次露出", ASSI) + CHECK_CLO("Ｖ露出", ASSI) && RAND:3 == 0
			;面部騎乗
			IF MAX(ABL:ASSI:施虐属性, ABL:MASTER:受虐属性) >= 3 && RAND:2
				IF CHECK_CLO("Ｖ露出", ASSI)
					PRINTFORMW 当%CALLNAME:MASTER%按照命令躺下后、%CALLNAME:ASSI%弯下腰将裸露出来的阴部压在了脸上！
				ELSEIF RAND:2
					PRINTFORMW 当%CALLNAME:MASTER%按照命令躺下后、%CALLNAME:ASSI%弯下腰坐在了脸上！
				ELSE
					IF RAND:2
						PRINTFORMW %CALLNAME:ASSI%将内衣脱下丢到一边、放下腰坐在了%CALLNAME:MASTER%的脸上！
					ELSE
						PRINTFORMW 当%CALLNAME:MASTER%按照命令躺下后、%CALLNAME:ASSI%脱掉了内衣放下腰坐在了脸上！
					ENDIF
					CALL SET_PANTIES, "", ASSI
					CALL SET_CEVENT, "助手で性処理：胖次下ろし", ASSI
				ENDIF
				TEQUIP:ASSI:面部騎乗 = 1
				TEQUIP:ASSI:六九式 = 91
				TEQUIP:MASTER:六九式 = 1
			ELSE
				PRINTFORMW %CALLNAME:ASSI%用69的姿势压在%CALLNAME:MASTER%的身上、无防备地露出股间的同时用舌頭责弄着阴茎！
				TEQUIP:ASSI:六九式 = 91
				;悪戯の内容を決める
				IF CHECK_CLO("Ｖ露出", ASSI) == 0
					IF RAND:2
						TEQUIP:MASTER:六九式 = 7
					ELSE
						TEQUIP:MASTER:六九式 = 1
					ENDIF
				ELSEIF RAND:7 == 0 && CALCF("尿道径", ASSI) > SIZE("指の直径", MASTER)
					TEQUIP:MASTER:六九式 = 15
				ELSEIF RAND:6 == 0
					TEQUIP:MASTER:六九式 = 14
				ELSEIF RAND:5 == 0
					TEQUIP:MASTER:六九式 = 13
				ELSEIF RAND:4 == 0
					TEQUIP:MASTER:六九式 = 8
				ELSEIF RAND:3 == 0
					TEQUIP:MASTER:六九式 = 7
				ELSEIF RAND:2
					TEQUIP:MASTER:六九式 = 2
				ELSE
					TEQUIP:MASTER:六九式 = 1
				ENDIF
			ENDIF

			SELECTCOM = 99
		ELSE
			;服を脱いでの奉仕。胖次を脱掉or全裸
			IF ABL:ASSI:露出癖 >= 3 && CHECK_CLO("裸体", ASSI) == 0 && EQUIP:ASSI:胖次 && RAND:3 == 0
				IF EQUIP:ASSI:胖次 == 0 || (ABL:ASSI:露出癖 >= 5 && RAND:2)
					PRINTFORMW %CALLNAME:ASSI%脱掉了所有的衣物、整齐地整理好放在一边之后在%CALLNAME:MASTER%的面前跪下用舌頭奉侍起阴茎
					CALL SETFLAG, "扒成裸体", ASSI
					CALL SET_CEVENT, "助手で性処理：裸体", ASSI
				ELSE
					PRINTFORMW 将胖次脱掉一半挂在腿上的%CALLNAME:ASSI%は、%TEXTR("熟练地/自然地")%取出%CALLNAME:MASTER%的阴茎将嘴唇贴了上去
					CALL SET_PANTIES, "", ASSI
					CALL SET_CEVENT, "助手で性処理：胖次下ろし", ASSI
				ENDIF
			;深喉
			ELSEIF CHECK("喉姦", ASSI) >= 2 && SIZE("阴茎", MASTER) && RAND:2
				PRINTFORMW %CALLNAME:ASSI%跪在%CALLNAME:MASTER%股间前、将坚硬地屹立在眼前的阴茎直到根部为止%TEXTR("一口气/整根/全部")%含入口中
				TEQUIP:ASSI:深喉 = MASTER
			ELSE
				PRINTFORM %CALLNAME:ASSI%は
				IF RAND:3 == 0
					PRINTFORM 跪在%CALLNAME:MASTER%的股间前
				ELSEIF RAND:2
					PRINTFORM 用自然的动作将%CALLNAME:MASTER%的
				ELSE
					PRINTFORM 一边用舌頭舔舐着一边将%CALLNAME:MASTER%的
				ENDIF
				PRINTFORM 阴茎取出来、
				;包茎
				IF TALENT:MASTER:包茎 && RAND:3 == 0
					;真性包茎
					IF COND("真性包茎", MASTER)
						PRINTFORMW 咬住覆盖着的包皮%TEXTR("用舌頭顶在尖端上旋转/从微微打開的包皮中用舌頭进行按摩")%
					;假性包茎
					ELSE
						PRINTFORMW 将多余的皮肤剥下之后%TEXTR("用嘴含住了龟頭/用舌頭舔舐龟頭/亲吻了龟頭")%
					ENDIF
				ELSEIF TALENT:ASSI:恋慕 && RAND:2
					PRINTFORMW 怜爱地%TEXTR("亲吻/用脸颊磨蹭")%之后放入口中
				ELSE
					PRINTFORMW %TEXTR("用舌頭舔/舔舐")%的同时可爱地向上看着这边侍奉着
				ENDIF
			ENDIF
		ENDIF
	CASEELSE
		SWAP TARGET, ASSI
		TRYCALLFORM TRAIN_MESSAGE_COM{SELECTCOM}
		SWAP TARGET, ASSI
	ENDSELECT
ENDIF

;SIMULATE_TRAINを使うためにASSIをTARGETにする
;-------------------------------------------------------------------------------------------------
MEMO_TARGET = TARGET
MEMO_ASSI = ASSI
TARGET = ASSI

	;主人の射精や助手の絶頂のカウントに使用
	MEMO:0 = EXP:MASTER:射精経験
	MEMO:1 = EXP:絶頂経験

	;回数はまだ調整していない
	FOR LCOUNT, 0, 10
		CALL SIMULATE_TRAIN_COM, SELECTCOM, 1
		SIF SELECTCOM:5 >= 0
			CALL SIMULATE_TRAIN_COM, SELECTCOM:5, 1
		;基本はMASTERが射精するまで
		SIF EXP:MASTER:射精経験 > MEMO:0
			BREAK
	NEXT

	;経験や珠などの獲得量計算
	CALL SIMULATE_TRAIN_CALC

	;結果発表
	CALL SIMULATE_TRAIN_SHOW

TARGET = MEMO_TARGET
ASSI = MEMO_ASSI
;-------------------------------------------------------------------------------------------------

SIF CEVENT("助手で性処理：地の文スキップ", ASSI)
	GOTO AFTER_TEXT_1

IF TEQUIP:ASSI:面部騎乗
	IF RANDF(4) == 0 && ABL:ASSI:施虐属性 >= 3
		PRINTFORMW %CALLNAME:MASTER%被%CALLNAME:ASSI%的面部騎乗闻嗅着酸臭味的同时、就这样吮吸着阴部射精了…
	ELSEIF CHECK_CLO("胖次露出", ASSI)
		IF RANDF(2) == 0
			PRINTFORMW %CALLNAME:MASTER%隔着胖次闻嗅着%CALLNAME:ASSI%的阴部可以说是%TEXTS("气味")%的气味的同时、就这样吮吸着阴部射精了…
		ELSE
			PRINTFORMW %CALLNAME:MASTER%一边亲吻着%CALLNAME:ASSI%的被%TEXTS("气味")%的气味浸染的胖次的同时、就这样吮吸着阴部射精了…
		ENDIF
	ELSE
		IF RANDF(3) == 0
			PRINTFORMW %CALLNAME:MASTER%被%CALLNAME:ASSI%的阴部塞满了口鼻、就这样吮吸着阴部射精了…
		ELSEIF RANDF(3) == 1
			PRINTFORMW %CALLNAME:MASTER%的脸被%CALLNAME:ASSI%阴部摩擦的同时、就这样吮吸着阴部射精了…
		ELSE
			PRINTFORMW %CALLNAME:MASTER%被%CALLNAME:ASSI%的股间侵犯的同时、就这样吮吸着阴部射精了…
		ENDIF
	ENDIF
ELSEIF TEQUIP:ASSI:深喉
	PRINTFORMW %CALLNAME:MASTER%的阴茎被化为了性器的%CALLNAME:ASSI%的嘴所捕食、无法反抗地被榨取出了精液…
;以下はSELECTCOMのみの分岐
ELSE
	SELECTCASE SELECTCOM
	CASE 91
		PRINTFORMW %CALLNAME:MASTER%在跪坐在股间用嘴进处理的%CALLNAME:ASSI%喉中、大量地射精了…
	CASE 92
		PRINTFORMW %CALLNAME:MASTER%在用嘴亲吻着龟頭用胸部撸动肉棒的%CALLNAME:ASSI%的口中、大量地射精了…
	CASE 99
		PRINTFORM %CALLNAME:MASTER%
		SELECTCASE TEQUIP:MASTER:六九式
		CASE 1
			IF CHECK_CLO("Ｖ露出", ASSI)
				PRINTFORM 把%CALLNAME:ASSI%的%TEXTR("阴部的各个角落全部来回舔舐/阴部上用脸来回摩擦/用嘴玩弄着阴核")%的同时
			ELSE
				PRINTFORM 把%CALLNAME:ASSI%的%TEXTR("的阴部隔着内衣享受/内裤放在脸上深深地闻嗅/的淫核作为重点进行攻击将内衣弄得湿乎乎的")%的同时
			ENDIF
		CASE 2
			PRINTFORM 用%TEXTR("手指按摩着/手指玩弄着")%%CALLNAME:ASSI%的后穴的同时
		CASE 7
			IF CHECK_CLO("Ｖ露出", ASSI)
				LOCALS = 闻嗅%TEXTS("气味")%
				PRINTFORM 打開%CALLNAME:ASSI%的小穴%TEXTR("又合上进行着玩弄/进行观察/插入手指检查着里面的情况/进行检查", LOCALS)%的同时
			ELSE
				PRINTFORM 将%CALLNAME:ASSI%的%TEXTR("小穴隔着内衣玩弄着/的胖次拉扯着玩弄着/的淫核作为重点进行攻击将内衣弄得湿乎乎的")%的同时
			ENDIF
		CASE 8
			PRINTFORM 将手指插入%CALLNAME:ASSI%的小穴中%TEXTR("咕哩咕哩地戳弄/打開又闭上/執拗地刮擦膣壁")%的同时
		CASE 13
			LOCALS = 享受着散发着热气的性器的%TEXTS("气味")%
			PRINTFORM 打開%CALLNAME:ASSI%的小穴%TEXTR("又合上进行着玩弄/进行观察/插入手指检查着里面的情况/进行检查", LOCALS)%的同时
		CASE 14
			PRINTFORM 打開%CALLNAME:ASSI%的后穴%TEXTR("观察每个角落/确认着色素沉淀的情况/进行检查")%的同时
		CASE 15
			PRINTFORM 将手指插入%CALLNAME:ASSI%的尿道进进出出地玩弄的同时
		ENDSELECT
		PRINTFORMW 、明明是在吮吸阴部还是放任让精液吐了出来…
	CASE 140
		PRINTFORMW %CALLNAME:MASTER%は%CALLNAME:ASSI%の頭を掴んで勢いよく腰を前後させ、腰を突き出して咽頭を擦りあげながら喉奥に充分射精した…
	CASEELSE
		CALL SAMEN_SHOOT_MESSAGE, ASSI
	ENDSELECT
ENDIF

$AFTER_TEXT_1
CALL DEL_CEVENT, "助手で性処理：地の文スキップ", ASSI

;口上があるかどうかを行数で格子
MEMO_LINECOUNT = LINECOUNT

CALL KOJO_GUEST, "助手で性処理：事後", ASSI

;台詞が無いなら汎用台詞
IF LINECOUNT == MEMO_LINECOUNT
	;特に口上が用意されていないならおあずけフラグＯＮ
	CALL SET_CEVENT, "助手で性処理：おあずけ", ASSI
ENDIF

SIF CEVENT("助手で性処理：地の文スキップ", ASSI)
	GOTO AFTER_TEXT_2

$AFTER_TEXT_2
CALL DEL_CEVENT, "助手で性処理：地の文スキップ", ASSI

EXP:ASSI:助手経験 += 1
;欲求も少々満たされます
CALL CALC, "欲求不満解消", ASSI, 10

;着衣の乱れを直す
CALL SETFLAG, "衣装全脱ぎ", ASSI

VARSET LOCALS
SPLIT MEMO_CLO, ",", LOCALS

FOR LCOUNT, 0, 10
	EQUIP:ASSI:(LCOUNT + 40) = TOINT(LOCALS:(LCOUNT*2) )
	CSTR:ASSI:(LCOUNT + 40) = %LOCALS:(LCOUNT*2 + 1)%
NEXT

RETURN 1


code:
@SELL_VIDEO
#DIM LCOUNT
#DIM VIDEO_VALUE
#DIM VIDEO_FLAG_VALUE
#DIMS VIDEONAME
#DIMS VIDEOLABEL
#DIMS VIDEO
#DIMS VIDEO_FLAG

;調教時に録画したビデオの売却
;TODO 価格設定再考・状況による価格をプラスできるか(処女開通時など)

;録画してなければスキップ
SIF STR:ビデオ内容 == ""
	RETURN 0

STR:ビデオ标题フラグ =
VIDEO_VALUE = 0

VARSET LOCALS
SPLIT STR:ビデオ内容, "/", LOCALS

FOR LCOUNT, 0, 20
	SIF LOCALS:LCOUNT == ""
		BREAK

	;フラグ重置
	VIDEO_FLAG =
	VIDEO_FLAG_VALUE = 0

	SELECTCASE LOCALS:LCOUNT
	CASE "愛撫"
		VIDEO_FLAG_VALUE += 50
		VIDEO_FLAG = Ｂ愛撫/Ｃ愛撫/
	CASE "舐陰"
		VIDEO_FLAG_VALUE += 50
		VIDEO_FLAG = Ｃ愛撫舐陰/
	CASE "肛門愛撫"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = Ａ愛撫/
	CASE "自慰"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 露出自慰/
	CASE "舐陰莖"
		VIDEO_FLAG_VALUE += 200
		VIDEO_FLAG = Ｐ愛撫口交/
	CASE "胸愛撫"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = Ｂ愛撫/
	CASE "接吻"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = Ｍ愛撫接吻/
	CASE "阴部愛撫"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = Ｃ愛撫/
	CASE "指挿入"
		VIDEO_FLAG_VALUE += 150
		VIDEO_FLAG = Ｖ愛撫/
	CASE "舐肛"
		VIDEO_FLAG_VALUE += 200
		VIDEO_FLAG = Ａ愛撫/
	CASE "手淫"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = Ｐ愛撫/
	CASE "给对方乳交"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 乳交Ｐ愛撫/
	CASE "给对方足交"
		VIDEO_FLAG_VALUE += 300
		IF PENIS(TARGET)
			VIDEO_FLAG = Ｐ愛撫足交/
		ELSE
			VIDEO_FLAG = Ｃ愛撫足交/
		ENDIF
	CASE "扒開阴道"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = Ｖ露出/
	CASE "扒開菊穴"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = Ａ露出/
	CASE "尿道指挿入"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 尿道愛撫/

	CASE "跳蛋"
		VIDEO_FLAG_VALUE += 250
		VIDEO_FLAG = Ｃ器具/
	CASE "振動棒"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 拡張Ｖ器具/
	CASE "Eマッサージャ"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = Ｃ器具/
	CASE "肛用振動棒"
		VIDEO_FLAG_VALUE += 700
		VIDEO_FLAG = 拡張Ａ器具/
	CASE "阴蒂帽"
		VIDEO_FLAG_VALUE += 150
		VIDEO_FLAG = Ｃ器具/
	CASE "ニプル帽"
		VIDEO_FLAG_VALUE += 150
		VIDEO_FLAG = Ｂ器具/
	CASE "搾乳器"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 搾乳Ｂ器具/
	CASE "飛機杯"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = Ｐ器具/
	CASE "淋浴"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = 器具/
	CASE "跳蛋挿入"
		VIDEO_FLAG_VALUE += 200
		VIDEO_FLAG = Ｖ器具/

	CASE "正常位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 正常位Ｖ性交/
	CASE "後背位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 後背位Ｖ性交/
	CASE "屈曲位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 屈曲位Ｖ性交/
	CASE "対面座位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 対面座位Ｖ性交/
	CASE "背面座位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 背面座位Ｖ性交/
	CASE "騎乗位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 騎乗位Ｖ性交/
	CASE "挿入刺激Ｇ点"
		VIDEO_FLAG_VALUE += 900
		VIDEO_FLAG = 玩弄Ｇ点Ｖ性交/
	CASE "挿入子宮口玩弄"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 玩弄子宮口Ｖ性交/
	CASE "子宮挿入"
		VIDEO_FLAG_VALUE += 1500
		VIDEO_FLAG = 子宮挿入Ｖ性交/
	CASE "肛門正常位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 正常位Ａ性交/
	CASE "肛門後背位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 後背位Ａ性交/
	CASE "肛門屈曲位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 屈曲位Ａ性交/
	CASE "肛門対面座位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 対面座位Ａ性交/
	CASE "肛門背面座位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 背面座位Ａ性交/
	CASE "肛門騎乗位"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 騎乗位Ａ性交/

	CASE "エースオブスペード"
		VIDEO_FLAG_VALUE += 400
		VIDEO_FLAG = レズＣ性交/
	CASE "雙乳間抽插"
		VIDEO_FLAG_VALUE += 400
		VIDEO_FLAG = 乳交Ｂ性交/
	CASE "乳頭奸"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 乳頭奸Ｂ性交/
	CASE "尿道挿入"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 尿道性交/

	CASE "手淫"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 奉仕手淫攻め/
	CASE "口交"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 奉仕口交攻め/
	CASE "乳交"
		VIDEO_FLAG_VALUE += 650
		VIDEO_FLAG = 乳交奉仕Ｂ攻め/
	CASE "素股"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 奉仕素股Ｃ攻め/
	CASE "泡泡浴"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 奉仕泡泡浴Ｂ攻め/
	CASE "足交"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 嗜虐足交攻め/
	CASE "肛門奉仕"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 嗜虐肛門奉仕攻め/
	CASE "内褲冲"
		VIDEO_FLAG_VALUE += 600
		VIDEO_FLAG = ぱんつ奉仕手淫攻め/
	CASE "舐陰させる"
		VIDEO_FLAG_VALUE += 250
		VIDEO_FLAG = 奉仕舐陰攻め/
	CASE "六九式"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 奉仕六九式攻め/
		IF PENIS(TARGET)
			VIDEO_FLAG += "Ｐ愛撫口交/"
		ELSE
			VIDEO_FLAG += "Ｃ愛撫舐陰/"
		ENDIF
		IF PENIS(MASTER)
			VIDEO_FLAG += "奉仕口交攻め/"
		ELSE
			VIDEO_FLAG += "奉仕舐陰攻め/"
		ENDIF
	CASE "洗面奶"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 乳交奉仕Ｂ攻め/

	CASE "正常位奉仕"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆正常位Ｖ逆性交/
		ELSE
			VIDEO_FLAG = 逆正常位Ｖ假阳具攻め
		ENDIF
	CASE "後背位奉仕"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆後背位Ｖ逆性交/
		ELSE
			VIDEO_FLAG = 逆後背位Ｖ假阳具攻め
		ENDIF
	CASE "騎乗位侵犯"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆騎乗位Ｖ逆性交/
		ELSE
			VIDEO_FLAG = 逆騎乗位Ｖ假阳具攻め
		ENDIF

	CASE "Ａ正常位奉仕"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆正常位Ａ逆性交/
		ELSE
			VIDEO_FLAG = 逆正常位Ａ假阳具攻め
		ENDIF
	CASE "Ａ後背位奉仕"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆後背位Ａ逆性交/
		ELSE
			VIDEO_FLAG = 逆後背位Ａ假阳具攻め
		ENDIF
	CASE "Ａ騎乗位侵犯"
		VIDEO_FLAG_VALUE += 800
		IF PENIS(TARGET)
			VIDEO_FLAG = 逆騎乗位Ａ逆性交/
		ELSE
			VIDEO_FLAG = 逆騎乗位Ａ假阳具攻め
		ENDIF

	CASE "打屁股"
		VIDEO_FLAG_VALUE += 250
		VIDEO_FLAG = 被虐/
	CASE "鞭"
		VIDEO_FLAG_VALUE += 450
		VIDEO_FLAG = 被虐器具/
	CASE "針"
		VIDEO_FLAG_VALUE += 700
		VIDEO_FLAG = 被虐器具/
	CASE "眼罩"
		VIDEO_FLAG_VALUE += 150
		VIDEO_FLAG = 被虐器具/
	CASE "縄・拘束"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 緊縛被虐器具/
	CASE "口球"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = 被虐器具/
	CASE "浣腸器＋塞子"
		VIDEO_FLAG_VALUE += 1500
		VIDEO_FLAG = スカトロ被虐Ａ器具/
	CASE "拡張充气"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 被虐器具/
	CASE "让对方浴尿"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 放尿被虐/
	CASE "強制口交"
		VIDEO_FLAG_VALUE += 800
		VIDEO_FLAG = 被虐Ｍ性交/
	CASE "拳交"
		VIDEO_FLAG_VALUE += 1500
		VIDEO_FLAG = 拡張Ｖ拳交/
	CASE "肛門拳交"
		VIDEO_FLAG_VALUE += 1500
		VIDEO_FLAG = 拡張Ａ拳交/
	CASE "両穴拳交"
		VIDEO_FLAG_VALUE += 2500
		VIDEO_FLAG = 拡張Ｖ拳交/拡張Ａ拳交/
	CASE "放尿"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = スカトロ放尿露出/
	CASE "剃毛"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 剃毛被虐露出/
	CASE "精飲"
		VIDEO_FLAG_VALUE += 600
		VIDEO_FLAG = 被虐奉仕/
	CASE "穿环"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 被虐露出/
	CASE "尿道导管"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 被虐尿道器具/
	CASE "深喉"
		VIDEO_FLAG_VALUE += 1000
		VIDEO_FLAG = 被虐Ｍ性交/
	CASE "尿道探针"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 被虐拡張尿道器具/

	CASE "野外PLAY"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 野外露出/
	CASE "浴室PLAY"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = お浴室/
	CASE "新妻PLAY"
		VIDEO_FLAG_VALUE += 500
		VIDEO_FLAG = 新妻/
	CASE "被逆推"
		VIDEO_FLAG_VALUE += 300
		VIDEO_FLAG = 嗜虐推倒/
	CASE "潤滑液"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = 潤滑液/
	CASE "媚薬"
		VIDEO_FLAG_VALUE += 100
		VIDEO_FLAG = 媚薬/
	;CASE "利尿剤"
	;	VIDEO_FLAG_VALUE += 100
	;設定されてないもの
	CASEELSE
		VIDEO_FLAG_VALUE += 300
	ENDSELECT

	;価格に加算
	VIDEO_VALUE += VIDEO_FLAG_VALUE

	;ビデオ标题生成用フラグに追加
	STR:ビデオ标题フラグ += VIDEO_FLAG
NEXT

SIF TALENT:処女 == 1
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:感情缺乏
	TIMES VIDEO_VALUE, 0.70
SIF TALENT:害羞
	TIMES VIDEO_VALUE, 1.10
SIF TALENT:娇小
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:巨乳
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:扶她
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:動物耳
	TIMES VIDEO_VALUE, 1.10
;SIF GETBIT(TALENT:人気, 0)
;	TIMES VIDEO_VALUE, 1.30
;SIF GETBIT(TALENT:人気, 1)
;	TIMES VIDEO_VALUE, 1.20
SIF TALENT:謎之魅力
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:魅力
	TIMES VIDEO_VALUE, 1.30
SIF TALENT:母乳体質
	TIMES VIDEO_VALUE, 1.20
SIF TALENT:淫乱
	TIMES VIDEO_VALUE, 1.10

;特殊な撮影内容の格子
VARSET LOCALS
SPLIT STR:ビデオジャンル, "/", LOCALS

;STR:ビデオジャンルの被りを無くす
STR:ビデオジャンル =

FOR LCOUNT, 0, 100
	SELECTCASE LOCALS:LCOUNT
	CASE ""
		BREAK
	CASEELSE
		SIF STRCOUNT(STR:ビデオジャンル, @"\[%LOCALS:LCOUNT%\]")
			CONTINUE
	ENDSELECT

	STR:ビデオジャンル += @"\[%LOCALS:LCOUNT%\]"

	SELECTCASE LOCALS:LCOUNT
	;レアor大人気+1,000$
	CASE "処女喪失", "美少女", "爆乳"
		VIDEO_VALUE += 1000
	;準レアor人気+500$
	CASE "扶她", "飲尿", "超乳", "白虎", "ＡＴＭ", "局部アップ", "肛門性交"
		VIDEO_VALUE += 500
	;人を選ぶマニア系
	CASE "貧乳", "剛毛", "スカ", "尿道"
		VIDEO_VALUE += 100
	;其他
	CASEELSE
		VIDEO_VALUE += 300
	ENDSELECT
NEXT

;出演するほど価値が下がる！
;VIDEO_VALUE = MULTIPLY(VIDEO_VALUE, MAX(100 - CFLAG:ＡＶ出演経験/3*5, 50) )

PRINTL
PRINTFORM 本次拍撮的ＡＶ
IF STR:ビデオジャンル != ""
	PRINTL
	PRINTFORML 类型：%STR:ビデオジャンル%
ENDIF
;ビデオ売却選択肢フラグの格子
IF TALENT:MASTER:禁断的知識 >= 3 && COND("合意：ＡＶ撮影")
	PRINTFORML 、要使用【認識阻害】使得主角不被认出然后出售吗？
	PRINTL [0] 是
	PRINTL [1] 否
	CALL INPUT_SELECT, 2
	IF RESULT == 0
		;売却価格で展開を変えるのもいいな
		IF VIDEO_VALUE >= 2000
			PRINTFORMW %CALLNAME:TARGET%主演的ＡＶ卖掉了、入手%MONEY_C(VIDEO_VALUE)%。
			MONEY += VIDEO_VALUE
			PRINTL
			CALL KOJO_EVENT, "ビデオ売却"
		ELSE
			PRINTFORMW %CALLNAME:TARGET%主演的ＡＶ几乎没有卖出去、仅有的一点利润消耗在了社团经费中。
			PRINTL
			CALL KOJO_EVENT, "ビデオ売却失敗"
		ENDIF
		;ビデオについての記録
		EXP:ＡＶ出演経験 += 1
		CFLAG:ビデオ売上合計 +=	VIDEO_VALUE
	ENDIF
ELSE
	PRINTFORMW 嘛……反正出售是不可能的、于是又看了一遍之后就把视频销毁了。
	CALL PRINT_STRW, "黄色_※ＡＶ的販売需要禁断的知識[認識阻害]、而且[ＡＶ撮影]的合意也是必要的※"
ENDIF


code:
;-------------------------------------------------
;この角色独自の衣装を設定する。制服、体操服、泳装は既存のものから選ぶのが楽です
;-------------------------------------------------
@CLOTHES_K0, ARGS
#DIMS MEMO_CLOMASTER

MEMO_CLOMASTER =
SELECTCASE ARGS
CASE "内衣"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_PANTIES, "四角褲/黒"
	ELSE
		CALL SET_PANTIES, "內褲"
		CALL SET_BRA, "简约胸罩"
	ENDIF
CASE "飾品"
	CALL SET_OTHER, SAVESTR:ARGS
CASE "制服"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_SHIRT, "男子西装/茶色"
		CALL SET_OUTER, "休闲褲/黒"
		CALL SET_SOCKS, "襪子"
		CALL SET_SHOES, "運動鞋"
	ELSE
		CALL SET_COSTUME, "西装外衣"
	ENDIF
CASE "体操服"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_COSTUME, "体操服运动衫"
	ELSE
		CALL SET_COSTUME, "体操服布鲁马"
	ENDIF
CASE "泳装"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_COSTUME, "衝浪胖次"
	ELSE
		CALL SET_COSTUME, "競泳泳装"
	ENDIF
CASE "便装"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_SHIRT, "男子西装/茶色"
		CALL SET_OUTER, "休闲褲/黒"
		CALL SET_SOCKS, "襪子"
		CALL SET_SHOES, "運動鞋"
	ELSE
		CALL SET_COSTUME, "西装外衣"
	ENDIF
CASE "居家服"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF SAVESTR:便装 != ""
		MEMO_CLOMASTER = %SAVESTR:便装%
	ELSEIF TALENT:男性
		CALL SET_SHIRT, "男子西装/茶色"
		CALL SET_OUTER, "休闲褲/黒"
		CALL SET_SOCKS, "襪子"
		CALL SET_SHOES, "運動鞋"
	ELSE
		CALL SET_COSTUME, "西装外衣"
	ENDIF
CASE "外出装"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF TALENT:男性
		CALL SET_SHIRT, "男子西装/茶色"
		CALL SET_OUTER, "休闲褲/黒"
		CALL SET_SOCKS, "襪子"
		CALL SET_SHOES, "運動鞋"
	ELSE
		CALL SET_COSTUME, "西装外衣"
	ENDIF
CASE "约会衣装"
	IF SAVESTR:ARGS != ""
		MEMO_CLOMASTER = %SAVESTR:ARGS%
	ELSEIF SAVESTR:外出装 != ""
		MEMO_CLOMASTER = %SAVESTR:外出装%
	ELSEIF TALENT:男性
		CALL SET_SHIRT, "男子西装/茶色"
		CALL SET_OUTER, "休闲褲/黒"
		CALL SET_SOCKS, "襪子"
		CALL SET_SHOES, "運動鞋"
	ELSE
		CALL SET_COSTUME, "西装外衣"
	ENDIF
ENDSELECT

IF MEMO_CLOMASTER != ""
	STR:換装衣装 = %MEMO_CLOMASTER%
	CALL SETFLAG, "衣装の換装", TARGET
ENDIF

path: ERB/EVENT_K/00 あなた/EVENT_K0_CLOTHES.ERB